// File: contracts/NewFeeCollector.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.1;
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/Address.sol";

contract NewFeeCollector is Ownable {
    uint256 public hsiMarketPlaceCollection;
    uint256 public hexartMarketPlaceCollection;
    uint256 public hexartArtistFeeCollection;

    enum CollectorTypes {
        BUYBURN,
        BUYDISTRIBUTE,
        HEXMARKET,
        HEDRONFLOW,
        BONUS
    }

    struct FeesCollectors {
        address payable feeAddress;
        uint256 share;
        uint256 hexArtShare;
        uint256 artistShare;
        uint256 amount;
    }

    mapping(uint256 => FeesCollectors) public feeMap;
    mapping(address => bool) public whitelistedAddresses;

    modifier isWhitelisted(address _address) {
        require(whitelistedAddresses[_address], "Not have permission");
        _;
    }

    receive() external payable {}

    /**
     *@notice Set Fee collector wallet details
     *@param feeType COLLECTORTYPES(enum)
     *@param wallet address payable
     *@param share uint256
     */
    function setFees(
        CollectorTypes feeType,
        address payable wallet,
        uint256 share,
        uint256 _hexArtShare,
        uint256 _artistShare
    ) external onlyOwner {
        require(wallet != address(0), "Zero address not allowed");
        require(share != 0, "Share must be greater than 0.");
        require(
            uint256(feeType) >= 0 && uint256(feeType) < 5,
            "Invalid fee type"
        );

        feeMap[uint256(feeType)] = FeesCollectors({
            feeAddress: wallet,
            share: share,
            hexArtShare: _hexArtShare,
            artistShare: _artistShare,
            amount: 0
        });
    }

    /**
     *@notice Update Fee collector wallet address and share
     *@param feeType COLLECTORTYPES(enum)
     *@param wallet address payable
     *@param share uint256
     */
    function updateFees(
        CollectorTypes feeType,
        address payable wallet,
        uint256 share,
        uint256 _hexArtShare,
        uint256 _artistShare
    ) external onlyOwner {
        require(wallet != address(0), "Zero address not allowed");
        require(share != 0, "Share must be greater than 0.");
        require(
            uint256(feeType) >= 0 && uint256(feeType) < 5,
            "Invalid fee type"
        );
        feeMap[uint256(feeType)] = FeesCollectors({
            feeAddress: wallet,
            share: share,
            hexArtShare: _hexArtShare,
            artistShare: _artistShare,
            amount: feeMap[uint256(feeType)].amount
        });
    }

    /**
     *@notice Assigns fees amount to fee collector structs
     *@param  value, buying amount for NFT, recieved from marketplace
     *@param  addShare, total fees share amount for NFT, recieved from marketplace
     */
    function manageFees(uint256 value, uint256 addShare)
        external
        isWhitelisted(msg.sender)
    {
        for (uint256 i = 0; i < 5; i++) {
            uint256 shareAmount = updateAmount(i, value, addShare);
            addShare = addShare - shareAmount;
        }
    }

    /**
     *@notice Assigns fees amount to fee collector structs
     *@param  value, buying amount for hexart, recieved from hex marketplace
     */
    function manageHexArtFees(uint256 value)
        external
        isWhitelisted(msg.sender)
    {
        require(value > 0, "value should not be 0");
        for (uint256 i = 0; i < 5; i++) {
            updateHexArtAmount(i, value);
        }
    }

    /**
     *@notice Assigns fees amount to fee collector structs
     *@param  value, artistFee , recieved from hex marketplace
     */
    function manageArtistFees(uint256 value)
        external
        isWhitelisted(msg.sender)
    {
        require(value > 0, "value should not be 0");
        for (uint256 i = 0; i < 5; i++) {
            updateArtistAmount(i, value);
        }
    }

    /**
     *@notice Claim Hexmarket amount
     */
    function claimHexmarket() external {
        uint256 id = uint256(CollectorTypes.HEXMARKET);
        claimBalances(id);
        claimHedronFlow();
    }

    /**
     *@notice Claim Bonus amount
     */
    function claimBonus() external {
        uint256 id = uint256(CollectorTypes.BONUS);
        claimBalances(id);
    }

    /**
     *@notice Claim HedronFlow amount
     */
    function claimHedronFlow() public {
        uint256 id = uint256(CollectorTypes.HEDRONFLOW);
        claimBalances(id);
    }

    /**
     *@notice Claim Buy and Burn amount
     */
    function claimBuyBurn() external {
        uint256 id = uint256(CollectorTypes.BUYBURN);
        claimBalances(id);
    }

    /**
     *@notice Claim Buy and distribute  amount.
     */
    function claimBuyDistribute() external {
        uint256 id = uint256(CollectorTypes.BUYDISTRIBUTE);
        claimBalances(id);
    }

    /**
     *  @notice Withdraw the extra eth available after distribution.
     */
    function withdrawDust() public onlyOwner {
        uint256 withdrawableAmount;
        uint256 nonWithdrawableAmount;
        for (uint256 i = 0; i < 5; i++) {
            nonWithdrawableAmount += feeMap[i].amount;
        }
        withdrawableAmount = address(this).balance - nonWithdrawableAmount;
        require(withdrawableAmount > 0, "No extra ETH is available");
        payable(msg.sender).transfer(withdrawableAmount);
    }

    /**
    @notice Add asset address to the whitelist.
    @param _addressToWhitelist, Address to whitelist.
    */
    function addAssetToWhitelist(address _addressToWhitelist) public onlyOwner {
        require(
            Address.isContract(_addressToWhitelist),
            "Only contract address are allowed"
        );
        whitelistedAddresses[_addressToWhitelist] = true;
    }

    /**
    @notice Remove asset address to the whitelist.
    @param _addressToWhitelist, Address to whitelist.
    */
    function removeAssetFromWhitelist(address _addressToWhitelist)
        public
        onlyOwner
    {
        whitelistedAddresses[_addressToWhitelist] = false;
    }

    /**
     *@notice  Get balance of this contract.
     *@return uint
     */
    function getBalance() internal view returns (uint256) {
        return address(this).balance;
    }

    /**
     *@notice Update amount to fee collector structs used by manageFees function
     *@param   id, Index of COLLECTORTYPES
     *@param  value, buying amount for NFT, recieved from marketplace
     *@param  addShare, total fees share amount for NFT, recieved from marketplace
     */
    function updateAmount(
        uint256 id,
        uint256 value,
        uint256 addShare
    ) internal returns (uint256) {
        uint256 shareAmount = (value * feeMap[id].share) / 1000000;
        if (shareAmount <= addShare) {
            feeMap[id].amount = feeMap[id].amount + shareAmount;
            hsiMarketPlaceCollection = hsiMarketPlaceCollection + shareAmount;
        } else {
            feeMap[id].amount = feeMap[id].amount + addShare;
            hsiMarketPlaceCollection = hsiMarketPlaceCollection + addShare;
        }

        return shareAmount;
    }

    /**
     *@notice Update amount to fee collector structs used by manageFees function
     *@param   id, Index of COLLECTORTYPES
     *@param  value, buying amount for NFT, recieved from marketplace
     */
    function updateHexArtAmount(uint256 id, uint256 value)
        internal
        returns (uint256)
    {
        uint256 shareAmount = (value * feeMap[id].hexArtShare) / 1000000;

        feeMap[id].amount = feeMap[id].amount + shareAmount;
        hexartMarketPlaceCollection = hexartMarketPlaceCollection + shareAmount;

        return shareAmount;
    }

    /**
     *@notice Update amount to fee collector structs used by manageFees function
     *@param   id, Index of COLLECTORTYPES
     *@param  value, buying amount for NFT, recieved from marketplace
     */
    function updateArtistAmount(uint256 id, uint256 value)
        internal
        returns (uint256)
    {
        uint256 shareAmount = (value * feeMap[id].artistShare) / 100;

        feeMap[id].amount = feeMap[id].amount + shareAmount;
        hexartArtistFeeCollection = hexartArtistFeeCollection + shareAmount;

        return shareAmount;
    }

    /** 
     @notice Claim Balance for the type of COLLECTORTYPES
     *@param  id, Index of COLLECTORTYPES
    */
    function claimBalances(uint256 id) internal {
        uint256 totalAmount = feeMap[id].amount;
        require(
            totalAmount <= getBalance() && totalAmount > 0,
            "Not enough balance to claim"
        );
        feeMap[id].amount = 0;
        feeMap[id].feeAddress.transfer(totalAmount);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}